

Modellierung von Photosynthese

1 Einleitung

Pflanzen sind in der Lage die Strahlungsenergie der Sonne zu nutzen, um einfache Moleküle – Kohlendioxid und Wasser – in komplexe organische Moleküle umzuwandeln, welche für Pflanzen und Tiere gleichermaßen als Energiequelle und molekulare Bausteine dienen. Außerdem wird durch Photosynthese Sauerstoff in die Luft freigesetzt, und gerade dieser Sauerstoff spielt eine wichtige Rolle für die zelluläre Atmung und die Bereitstellung von ATP. Pflanzen, Tiere und Menschen würden ohne die Photosynthese ersticken und verhungern.

Nach. RAVEN, H. et al. (2000).

Photosynthese ist also einer der wichtigsten, wenn nicht der wichtigste biologische Prozess auf der Erde und es liegt nahe sich intensiv mit ihm zu beschäftigen und die Gesetzmäßigkeiten die ihn beeinflussen zu untersuchen.

2 Theorie

Die biologische Funktionsweise der Photosynthese auf Blattebene ist gut bekannt. THORNLEY beschreibt das Sättigungsverhalten der Brutto-Photosynthese GPR in Abhängigkeit von der Photonenflussdichte I mit einer nicht rechtwinkligen Hyperbel (Gl. 1)

$$GPR = \frac{(\alpha \cdot I + P_{\max}) - \sqrt{(\alpha \cdot I + P_{\max})^2 - 4\alpha \cdot I \cdot P_{\max} \cdot \theta}}{2\theta} \quad (\text{Gl. 1})$$

Wobei: α = Lichtausnutzungseffizienz, P_{\max} = maximale Photosyntheserate und θ = Parameter für die Krümmung der Hyperbel

Die Nettophotosynthese NPR erhält man, wenn man von GPR die Atmungsrate R_d abzieht (Gl. 2).

$$NPR = GPR - R_d \quad (\text{Gl. 2})$$

Für die Dunkelatmungsrate kann bei 20° C für Sonnennadeln z.B. 1,6 für Schattennadeln 0,8 $\mu\text{mol CO}_2/(\text{m}^2 \cdot \text{s})$ angesetzt werden.

Das Problem dabei die Photosynthese für einen ganzen Bestand und nicht nur für ein Blatt zu berechnen, ist die Simulation der Strahlungsverteilung im Bestand. Strahlung wird von den obersten Blättern absorbiert, reflektiert und transmittiert. Um die Photosynthese von unteren Blattschichten zu berechnen, muss bekannt sein, wie groß die Photonenflussdichte der Strahlung ist, die sie noch erreicht.

Die Berücksichtigung all dieser Einflussgrößen, würde das Modell extrem kompliziert machen. Eine starke Vereinfachung, die aber als geeignete Näherung erscheint ist die, die Blätter als schwarze Körper ohne Reflexion und Transmission aufzufassen.

Nun gilt es den Extinktionskoeffizienten zu ermitteln, mit welchem die Strahlung abnimmt, sobald sie tiefere Blattschichten erreicht. Nimmt man an, dass die Blätter eines Baumes sowohl in azimutaler, als auch in horizontaler Richtung gleich verteilt sind, kann man sich vorstellen alle Blätter in Form einer Kugel anzuordnen. Mit dieser Annahme und dem Lambert und Beer'schen Gesetz, kann der Extinktionskoeffizient bei bekannter Blattoberfläche (LAI) berechnet werden. Er ist außerdem abhängig vom Sonnenwinkel β . Die geometrischen Zusammenhänge sind in Abbildung 1 dargestellt.




Abbildung 1: Projektion der Kugeloberfläche senkrecht zur Strahlungsrichtung (A_p) und auf die Horizontale (A_h) aus Monteith und Unsworth (1990).

Mit den gemachten Vereinfachungen zu den Strahlungseigenschaften der Blätter und der Strahlungsverteilung im Kronenraum, kann nun die Photosynthese für einzelne Schichten modelliert werden, wobei jeweils die kumulierte Abschattung der oberen Schichten berücksichtigt wird.

2.1 SAS-Code

Die oben gemachten Annahmen resultieren in Folgendem SAS-Programm (Skript 1).

Skript 1: Verwendeter SAS-Code

```

/* Macro */

%macro beer_PS(oset,lai,beta,F_PAR_b, F_PAR_d);
  /*
  Oset: Ausgabedatensatz
  beta: Sonnenhöhenwinkel in °
  FPar_b: direkte PAR-Flussdichte  $\mu\text{E}/(\text{m}^2\text{s})$ 
  FPar_d: diffuse PAR-Flussdichte  $\mu\text{E}/(\text{m}^2\text{s})$ 
  LAI: Blattflächenindex
  */
  DATA &oset;
  retain sum_pl; /* die Photosyntheseraten werden über die
verschiedene Blattwinkelklassen einer Schicht aufsummiert */
  retain sum_NPc 0 sum_Pc 0 sum_RD 0; /* Photosyntheseraten, die
kumulativ über die Blattfläche
                                aller
                                Schichten
aufsummiert werden. Initialisierung mit 0 */
  /* Konstanten: */
  pi = 4*atan(1);

  /* Photosyntheseparameter: Leverenz & Jarvis, 1981 Sonnennadeln
der Sitkafichte */
  theta=0.7;
  alpha=0.048;
  Rd = 1.6; /*Dunkel-Respiration CO2-Abgabe pro m2 Blattfläche */
  p_max=12; /* Sättigungswert der Brutto-Photosynthese */

  /* Randbedingungen */
  Lai=&LAI;/* Blattflächenindex m2 proj. Blattfläche / m2 Grundfläche
*/
  beta=&beta*pi/180;/* im Bogenmaß Sonneninkinationswinkel bezogen auf
die Grundfläche */
  sinb = sin(beta);
  f_par_b0=&F_PAR_b; /* direkte Einstrahlung außerhalb des
Kronenraumes b für "beam"*/
  f_par_d0=&F_PAR_d; /* diffuse Einstrahlung außerhalb des
Kronenraumes */

  fl=0.1; /* Blattflaeche pro Schicht in m2*/

  /*Extinktionskoeffizient für diffuse Strahlung */
  k_d = 0.75;
  /*Extinktionskoeffizient für direkte Strahlung */
  k_b = 0.5*(1/sinb);

  /* Schleife über (LAI-1)/fl Kronenschichten */
  DO L=fl/2 TO (lai-fl) BY fl; /* L ist die kumulative Blattfläche
über der betrachteten Schicht */
  /* Lambert Beer'sches Gesetz:
F_par ist die über die Grundfläche gemittelte PAR
nach Passage einer einer kumulativen Blattfläche von L.
Die Extinktion (=exp(-0.5*(1/sin(beta))*1))
sagt uns, dass die Schattenfläche im Vergleich zur
Grundfläche entsprechend mit zunehmendem L größer wird.
F_par ist tatsächlich ein Mittelwert, dem Modell zufolge

```

```

        ist es aber so, dass wir in einem Teil des Querschnittes
        totalen Schatten haben und in dem Rest die volle
        Sonneneinstrahlung. Dies ist wichtig zu unterscheiden,
wenn die Photosynthese
        berechnet werden soll, da diese nicht-linear von der
Strahlung abhängt.
        */

        /* Diffuse Strahlung */
        F_PAR_d= f_par_d0*exp(-k_d*1);/* Diffuse Strahlung
Annahme uniform overcast sky */
        /* Brutto-Photosynthese der Blätter, die nur diffuse
Strahlung erhalten */
        a=f_par_d*alpha+p_max; /* Hilfsvariable */
        gpr_d=(a-sqrt(a**2-4*f_par_d*alpha*p_max*theta))/(2*theta);
/*µmol CO2/(m²s)*/

        f_par_b=f_par_b0 * exp(-k_b*1);/* gemittelte direkte PAR-
Flussdichte µE/(m²s)*/
        f_b=fl*exp(-k_b*1); /* Blattfläche einer Schicht, die direkt
bestrahlt ist */

        /* Initialisierung der Summenvariablen für die Schleife
über die Blattwinkel */
        sum_pl=0;

        /* Schleife über die Blattwinkel (beta_l), da die Intensität der
Strahlung von beta_l abhängt,
        mit dem sie auf die Blätter auftrifft. Es gilt: F_par =
F_par0 * sin(beta_l) */
        DO beta_l = Pi/80 to Pi/2 by Pi/40 ;
        /* direkte Strahlung Photonenflussdichte in der
Winkelklasse */
                sin_delta= sin(abs(beta_l-beta));/* Sinus des
Winkels zwischen Strahlung und Blattoberfläche*/
                f_par=f_par_b0*sin_delta+ F_PAR_d; /* PAR-Flussdichte auf
dem Blatt */
                /* Lichtantwortfunktion: Nicht rechtwinklige Hyperbel,
Thornley, 1976 */
                a=f_par*alpha+p_max; /* Hilfsvariable */
                gpr_b=(a-sqrt(a**2-4*f_par*alpha*p_max*theta))/(2*theta);
/*Brutto-Photosynthese pro m**2 Blattfläche */
                /* Berechnung der Photosynthese pro Schicht und
Winkelklasse */
                GPR_b=GPR_B*f_b/20;/*Brutto-Photosyn. pro 1/20 bestrahlter
Fläche :1/20 wg. 20 Winkelklassen mit je 1/20 der Blattfläche
der sonnenbeschienenen Blätter */
                sum_pl=sum_pl+GPR_b;
        end;

        /* Raten für die jeweilige Schicht */
        GPR_B = sum_pl;/* Bruttphotosynthese der
sonnenbeschienenen Blätter */
        GPR_d=GPR_d*FL*(1-f_b/FL);/* Bruttphotosynthese der
beschatteten Blätter */
        GPR = GPR_b + GPR_d; /* Bruttphotosynthese */
        F_Par=F_PAR_b + F_PAR_d; /* PAR-Flussdichte */
        NPR = GPR - Rd * FL;/* Nettophotosynthese, Blätter betreiben
mit unterschiedlicher Rate Photosynthese,
        alle Blätter der Schicht
atmen mit der selben Rate. */
        RDr=-RD*FL; /* Dunkelatmung */

```

```

        /* kumulative Raten bis zur Schicht*/
sum_GPR + GPR;
sum_NPR + NPR;
sum_RD  + (-RD*FL);

        /* Berechnung der relativen PAR-Absorption nach L */
relPAR_d = 1-(F_PAR_d/f_par_d0);
relPAR_b = 1-(F_PAR_b/f_par_b0);

OUTPUT;
END;
drop pi a sum_pl beta_l sin_delta ;
run;
%mend;

%beer_PS(beer_PS,4,51,600,200);
proc insight data=beer_PS;
  line GPR NPR Rdr*1;
  line GPR GPR_b GPR_d *1;
  line F_PAR F_par_b F_PAR_d relPAR_d * 1;
  line sum_GPR sum_NPR sum_RD * 1;
  line relpar_d relpar_b * L;
run;
quit;

%beer_PS(beer_PS_2,4,51,800,0);

/* GIF-Grafik erzeugen */
GOPTIONS DEVICE=gif733
          noborder
          GACCESS='c:\studium\forst\master\3poek\ueb8\ueb8_00_01.gif'
          GSFMODE=REPLACE
          FTEXT=complex
          HTEXT=1.2;
%let color1=red;
%let color2=blue;
%let color3=green;
%let color4=purple;
%let color5=red;
%let color6=blue;
%let color7=green;
%let color8=purple;

symbol1 i = join c = red w=2 l=1;
symbol2 i = join c = blue w=2 l=1;
symbol3 i = join c = green w=2 l=1;
symbol4 i = join c = black w=2 l=1;

axis1 order = 0 to 4 by 1 minor=(n=2) label=("L (kumulative Blattflaeche ueber Schicht) ");
axis2 label=(angle = 90 "(10^-6) mol CO2 / ((m^2)*s)");
axis3 label=(angle = 90 "(10^-6) E / ((m^2)*s)");
legend1 label=none value=("GPR" "NPR" "Rdr");
legend2 label=none value=("GPR" "GPR_b" "GPR_d");
legend3 label=none value=("F_PAR" "F_PAR_b" "F_PAR_d");
legend4 label=none value=("sum_GPR" "sum_NPR" "sum_RD");
proc gplot data = beer_ps;
  plot (GPR NPR Rdr) * L /frame overlay haxis = axis1 vaxis = axis2
  legend=legend1;

```

```

    plot (GPR GPR_b GPR_d) * L /frame overlay haxis = axis1 vaxis =
axis2 legend=legend2;
    plot (F_PAR F_PAR_b F_PAR_d) * L /frame overlay haxis = axis1 vaxis
= axis3 legend=legend3;

    plot (sum_GPR sum_NPR sum_RD) * L /frame overlay haxis = axis1
vaxis = axis2 legend=legend4;
run;quit;

/* Grafik ohne diffuse Strahlung */
proc gplot data = beer_ps_2;
    plot (GPR NPR Rdr) * L /frame overlay haxis = axis1 vaxis = axis2
legend=legend1;
run;quit;

```

Mit dem erstellten Makro können leicht verschiedene Szenarien berechnet werden, wobei jeweils der Blattflächenindex (LAI), der Sonnenhöhenwinkel (β) und die direkte und die diffuse PAR-Flussdichte variiert werden können.

3 Ergebnisse

Mithilfe der Lichtantwortfunktion nach THORNLEY (1976) wurde für folgende Ausgangsparameter die Photosyntheseaktivität berechnet:

Schichtdicke:	$FL = 0,1 \text{ m}^2/\text{m}^2$
Blattflächenindex:	$LAI = 4 \text{ m}^2/\text{m}^2$
Inklinationswinkel:	$\beta = 51^\circ$
Flussdichte der direkten Strahlung:	$F_{PAR_b} = 600 \text{ } \mu\text{E}/\text{m}^2\text{s}$
Flussdichte der diffusen Strahlung:	$F_{PAR_d} = 200 \text{ } \mu\text{E}/\text{m}^2\text{s}$

Da die Photosyntheseaktivität vom Strahlungsangebot abhängt erscheint es sinnvoll zuerst die Strahlungsflussdichten in den einzelnen Schichten zu betrachten:

Abbildung 2 stellt die Flussdichte der direkten- (F_{PAR_b}) und der diffusen (F_{PAR_d}), Strahlung, sowie die aus der Summe von F_{PAR_b} und F_{PAR_d} berechnete Gesamtstrahlungsflussdichte (F_{PAR}) dar.

Es zeigt sich ein hyperbelförmiger Verlauf der Kurven, die Flussdichten nehmen dabei vom jeweiligen Niveau der Freilandstrahlung ausgehend mit zunehmender kumulativen Blattfläche über der betrachteten Schicht ab.




Abbildung 2: Strahlungsflussdichten in Abhängigkeit von der kumulativen Blattfläche über der Schicht

Ausgehend von den Strahlungsflussdichten und den Flächen von Sonnen- und Schattenblättern in der jeweiligen Schicht lässt sich die Bruttphotosyntheserate der sonnenbeschienenen Blätter (GPR_b) und der beschatteten Blätter (GPR_d) berechnen. Die Summe von GPR_b und GPR_d ergibt die gesamte Bruttphotosyntheserate (GPR).

Abb. 3 zeigt die drei Bruttphotosyntheseraten in Abhängigkeit von der kumulativen Blattfläche über der betrachteten Schicht.

Ähnlich wie die Kurven der Strahlungsflussdichten in Abb.1 zeigt auch die Kurve der GPR_b einen hyperbelförmigen Verlauf und nimmt mit zunehmender kumulativer Blattfläche über der betrachteten Schicht ab, was unmittelbar einleuchtend erscheint, da die Photosyntheserate von der Strahlungsflussdichte abhängt.

Die Kurve der GPR_d hingegen weist einen deutlich anderen Verlauf auf, in der obersten Schicht ($L=0$) ist keine Photosyntheseaktivität der Schattenblätter vorhanden, was daran liegt dass es keine Schattenblätter in dieser Schicht gibt.

Mit zunehmender kumulativer Blattfläche über der Schicht nimmt auch die GPR_d zu bis sie bei $L=1$ ihr Maximum erreicht. Dies liegt daran, dass zwar die PAR_d abnimmt, aber gleichzeitig die Fläche der Schattenblätter stärker zunimmt. Im weiteren Verlauf sinkt die GPR_d wieder ab, da nun die Abnahme der PAR_d die Zunahme der Fläche der Schattenblätter überwiegt.

Die Kurve der GPR zeigt hingegen, ähnlich wie die Kurven der Strahlungsflussdichten und der GPR_b , einen hyperbelförmigen Verlauf und nimmt mit zunehmender kumulativer

Blattfläche über der betrachteten Schicht ab. Dabei erscheint die Hyperbel jedoch deutlich gestreckter, da der steile Abfall in den oberen Schichten durch den in diesem Bereich gegenläufigen Verlauf der GPR_d Kurve gebremst wird.




Abbildung 3: Bruttphotosyntheseraten in den einzelnen Schichten

Nach Berechnung der Bruttphotosyntheserate (GPR) erscheint es interessant auch die Nettphotosyntheserate (NPR) zu berechnen, dieses erfolgt durch Subtraktion der Atmungsrate (Rdr) .

Abbildung 4 zeigt den Verlauf der GPR, NPR und Rdr in den einzelnen Schichten. Da im Modell die Annahme getroffen wurde Rdr sei für alle Blätter konstant, zeigt sich ein x-Achsen paralleler Verlauf von Rdr.

Die NPR-Kurve verläuft demnach um den Betrag von Rdr verschoben parallel zur GPR Kurve.




Abbildung 4: Photosynthese- und Respirationsraten in Abhängigkeit der kumulativen Blattfläche über der Schicht

Nachdem die Photosynthese und Respirationsraten für alle einzelnen Schichten berechnet wurden, stellt sich die Frage wie sich die Situation im Gesamtbestand, d.h. als Summe aller Schichten, darstellt.

Abbildung 5 zeigt die summierte Brutto- und Nettophotosyntheseraten (sum_GPR und sum_NPR) sowie die summierte Atmungsrate (sum_RD).

Die Kurve der sum_GPR weist einen typischen Sättigungs-Verlauf auf und nimmt bei steigender Blattfläche zu, bis sie sich einem Grenzwert nähert, an dem alle zur Verfügung stehende Strahlung genutzt würde.

Die Kurve der sum_RD weist hingegen einen linearen, abfallenden Verlauf auf, wird also mit zunehmender Blattfläche gleichmäßig negativer, da das Modell ja unterstellt alle Blätter würden gleich viel Atmung betreiben.




Abbildung 5: Summierte Photosyntheseraten und summierte Atmungsrate

Die sum_NPR lässt sich auch als Summe der sum_GPR und sum_RD berechnen, sie steigt demnach so lange an, wie die Grenzrate $d(\text{sum_GPR}) / d(L)$ größer als der Betrag der Grenzrate $d(\text{sum_RD}) / d(L)$ ist. Nach Erreichen dieses Maximums fällt sie dann ab und würde bei einer weiteren Steigerung von L annähernd parallel mit sum_RD verlaufen.

Am Maximum von sum_NPR ($L \approx 3,2$) liegt der optimale LAI für einen Bestand der vorgegebenen Parametrisierung, da jede zusätzliche Blattfläche einen Verlust an Nettophotosyntheseleistung bedeuten würde und somit nicht „rentabel“ wäre.

4 Diskussion

4.1 Auswirkungen der Schichtdicke auf die Ergebnisse der Simulation

Je größer die Schichtdicke ist, d.h. je größer die Blattoberfläche innerhalb einer Schicht ist, desto weniger Punkte werden zur Erstellung der Kurven berechnet, was zur Folge hat dass die Kurven eine eckige Form bekommen.

Dieses wäre an sich kein allzu großes Problem, im Gegensatz zu der Tatsache, dass die Annahme von dickeren Schichten zu einem größeren Diskretisierungsfehler führt. Die Annahme einer großen Blattfläche in der obersten Schicht führt zu der Annahme, dass eine große Blattfläche komplett unbeschattet ist und somit in den Genuss der maximalen Einstrahlung kommt. Dieses wiederum führt zur Schätzung von deutlich höheren Photosyntheseraten in der obersten Schicht. Gleichzeitig führt eine große unterste Schicht zu dem Ergebnis das eine große Blattfläche maximal beschattet wird und somit eine zu einer deutlich geringeren Photosyntheserate in dieser Schicht.

Abb. 6 stellt die Situation analog zu Abb. 4 bei einer angenommenen Blattfläche je Schicht von $0,5 \text{ m}^2/\text{m}^2$ und ansonsten unveränderten Ausgangsparametern dar (man beachte die unterschiedliche Skalierung der y-Achse)




Abbildung 6: Photosynthese- und Respirationsraten in Abhängigkeit der kumulativen Blattfläche über der Schicht, bei einer Blattfläche je Schicht von $0,5 \text{ m}^2/\text{m}^2$

Es wäre nun zu vermuten, dass sich Überschätzung der Photosyntheserate in den oberen Schichten und ihre Unterschätzung in den unteren Schicht gegenseitig wenigstens teilweise

aufheben, so dass die summierten Photosyntheseraten wieder ähnliche Werte wie bei der Berechnung mit einer kleineren Schichtdicke aufweisen.

Abb. 7 stellt die Situation analog zu Abb. 5 bei einer angenommenen Blattfläche je Schicht von $0,5 \text{ m}^2/\text{m}^2$ und ansonsten unveränderten Ausgangsparametern dar (man beachte die unterschiedliche Skalierung der y-Achse).




Abbildung 7: Summierte Photosyntheseraten und summierte Atmungsrate, bei einer Blattfläche je Schicht von $0,5 \text{ m}^2/\text{m}^2$

Entgegen den ersten Erwartungen tritt kein Aufhebungseffekt ein, stattdessen findet eine Verschiebung des Maximums von sum_NPR nach links statt ($L \approx 2,0$), es wird also ein geringerer optimaler LAI für den Bestand berechnet. Der Grund ist in dem steileren Anfangsverlauf von sum_GPR zu sehen, was zu einer stärkeren Krümmung der Kurve und somit dazu führt, dass die Grenzrate $d(\text{sum_GPR}) / d(L)$ früher gleich dem Betrag der Grenzrate $d(\text{sum_RD}) / d(L)$ wird.

4.2 Größe der relative PAR-Absorption in Abhängigkeit von LAI und

Sonnenwinkel für diffuse und für direkte Strahlung

Unter relativer PAR_Absorption versteht man den Anteil der Freilandstrahlung der von den Blättern absorbiert wird.

Es erscheint unmittelbar einleuchtend, dass eine große Blattfläche mehr Strahlung absorbieren kann als eine kleine Blattfläche, so dass davon auszugehen ist das bei einem größeren LAI mehr Strahlung absorbiert wird als bei einem kleineren.

Abb. 8 zeigt die Situation für $\beta=51^\circ$, es zeigt sich ein Sättigungsverlauf der Kurven, wobei auffällt dass die Kurven für die direkte- und indirekte PAR-Absorption annähernd gleich verlaufen.




Abbildung 8: Relative PAR-Absorption bei einem Sonnenwinkel von $\beta=51^\circ$

Eine Veränderung des Sonnenwinkels führt zu einer Verschiebung der Absorptionskurve der direkten Strahlung, die Absorptionskurve der diffusen Strahlung bleibt hingegen unverändert. Eine Erklärung hierfür ist dass der Extinktionskoeffizient der diffusen Strahlung zumindest im Modell als konstant angenommen wird, wohingegen der Extinktionskoeffizient der direkten Einstrahlung vom Inklinationwinkel abhängt. Da das Modell schwarze Blätter voraussetzt ist die Extinktion gleich der Absorption und somit die Absorption ebenfalls vom Inklinationwinle abhängig.

Abb. 9 stellt den Verlauf der relativen Absorption für einen sehr kleinen Sonnenwinkel ($\beta = 10^\circ$) dar, Abb. 10 hingegen für den Zenith des Sonnenwinkels ($\beta=90^\circ$) wie er nur in den Tropen erreicht wird dar.

Es zeigt sich, dass je flacher der Sonnenwinkel ist, desto steiler die Absorptionskurve der direkten PAR am Anfang ansteigt, also bereits bei einem geringen LAI relativ viel direkte Strahlung absorbiert wird, wohingegen bei einem steilen Sonnenwinkel auch große LAIs viel Licht durchlassen. Dieses erklärt warum tropische Wälder i.d.R. einen wesentlich höheren LAI aufweisen als temperate- oder boreale Wälder: Nur der steile Sonnenwinkel in den Tropen ermöglicht es, auch bei hohem LAI in den oberen Schichten, noch genug Strahlungsfluss in den unteren Schichten zu haben das die Nettophotosyntheserate positiv ist.




Abbildung 9: Relative PAR-Absorption bei einem Sonnenwinkel von $\beta=10^\circ$




Abbildung 10: Relative PAR-Absorption bei einem Sonnenwinkel von $\beta=90^\circ$

4.3 Beziehung zwischen LAI und Bruttophotosynthese

Steigt der LAI, steigt die summierte GPR, bis sich die Kurve asymptotisch einem Maximum nähert. Solange in den untersten Schichten noch Photosynthese betrieben wird steigert sich die summierte GPR um diesen Betrag. Da durch geringere Photonendichten in unteren

Schichten auch die Photosynthese geringer ist, steigt die summierte GPR immer langsamer. Die GPR in der jeweiligen Schicht sinkt mit zunehmender kumulierter Blattfläche über ihr und nähert sich null an (siehe Abbildungen 4 und 5).

4.4 Einfluss der diffusen Strahlung auf die Beziehung zwischen GPR und LAI

Abbildung 11 zeigt Brutto- und Nettophotosynthese und Atmung ohne diffuse Strahlung. Im Vergleich zu Abbildung 4 wird deutlich, dass die GPR nun einen etwas anderen Verlauf nimmt. Sie fällt steiler ab, da die puffernde Wirkung der diffusen Strahlung entfällt, die erst ihre Wirkung entfaltet, wenn beschattete Blätter vorhanden sind (s. Abbildung 3, GPR_d).




Abbildung 11: Photosynthese- und Respirationsraten in Abhängigkeit der kumulativen Blattfläche über der Schicht ohne diffuse Strahlung

4.5 Beziehung zwischen LAI und Nettophotosynthese

Es gelten die gleichen Beziehungen wie bei der GPR, mit dem Unterschied, dass sich die NPR der jeweiligern Schicht nicht asymptotisch der Null sondern der Atmungsrate nähert, also negativ werden kann. Dies passiert, wenn die Atmungsrate betragsmäßig über der GPR liegt. Das ist auch ein Grund dafür, dass Wälder nicht unendlich dicht belaubt sind. Einer marginalen Zunahme der summierten GPR würde eine starke Abnahme der summierten NPR gegenüberstehen.

4.6 Auswirkungen einer Klumpung der Blattfläche

Je höher der Klumpungsgrad, desto weniger Strahlung wird bei gleichem LAI vom Kronendach absorbiert, was zur Folge hat, dass mehr Strahlung zum Boden transmittiert wird. Vermutlich ist dann auch die Photosynthese geringer und das dargestellte Modell überschätzt sie, da dort keine Klumpung vorgesehen ist.

4.7 Änderungen wenn die Blätter nicht schwarz wären

Schwarze Blätter absorbieren alle ankommende Strahlung. Wären die Blätter nicht schwarz, würden sie einen Teil der Einstrahlung reflektieren, was zu einer Erhöhung der diffusen Strahlung führen würde. Gleichzeitig würde der Extinktionskoeffizient geringer, da durch nicht schwarze Blätter auch Transmission möglich wäre.

Die Erhöhung der diffusen Strahlungsintensität und der verminderte Extinktionskoeffizient hätten eine Steigerung der Photosyntheseaktivität in den unteren Schichten zur Folge, da hier das Strahlungsangebot deutlich verbessert würde.

In den oberen Schichten bliebe die Photosyntheseaktivität konstant, solange davon ausgegangen werden kann, dass die Reflektionsrate der Blätter nicht zu groß ist und sie somit nicht in eine Mangelsituation geraten. Dies wäre der Fall wenn die Strahlung, die eigentlich im Palisadenparenchym für die Photosynthese gebraucht würde, also absorbiert werden müsste, stattdessen reflektiert würde.

4.8 Zusammenfassung der problematischen Vereinfachungen des Vorgehens

Wie jedes Modell, ist auch das Modell von THORNLEY eine vereinfachte Abbildung der Wirklichkeit. Die Vereinfachungen der Vorgehensweise sind daher nicht als Mangel zu bewerten, aber bei der Interpretation insoweit zu berücksichtigen, dass man sich immer vergegenwärtigt das es sich bei den getroffenen Aussagen um Abschätzungen und nicht etwa um exakte, die Realität zu 100% richtig abbildende Ergebnisse handelt.

Insofern ist trotz der folgenden recht langen Liste mit problematischen Vereinfachungen das Modell als solches durchaus als brauchbar zu bewerten.

- Reale Blätter sind nicht schwarz
- Blätter in einem realen Wald sind geklumpt
- Die Gesamtmenge der Blätter lässt sich nicht zwangsläufig auf eine Kugel projizieren

- Eine Beschattung der Blätter durch andere Organe der Bäume (Zweige, Äste, Stämme) oder durch Eigenheiten des Reliefs (Berge) wird nicht abgebildet
- Eine Halbschattensituation ist nicht vorgesehen
- Die diffuse Strahlung ist in der Realität nicht völlig gleichmäßig
- Physiologische Unterschiede von Sonnen- und Schattenblättern etwa in der Anzahl Reihen des Pallisadenparenchyms und die daraus resultierenden unterschiedlichen Photosyntheseraten bei gleichen Lichtverhältnissen werden nicht abgebildet.
- Die Photosyntheserate hängt nicht nur mit dem Strahlungsangebot zusammen, auch andere Faktoren wie etwa Wasserversorgung oder Temperatur spielen hierbei eine wichtige Rolle.
- Die Atmungsrate der Blätter ist in der Realität nicht konstant.

5 Literatur

Raven, P., Evert R., Eichhorn S., 2000: Biologie der Pflanzen, S. 137, Verlag De Gruyter, Berlin, 1032 Seiten